
Abstract. We present the Maude Termination Assistant (MTA). Its
purpose is to complement the fully automatic Maude Termination Tool
(MTT) [2] in two ways: (i) to help users find a proof when MTT may
not find one; and (ii) for teaching purposes. The main features supported
by MTA are: (1) termination proofs of order-sorted equational theories,
which may include conditional rules without extra variables; (2) such
proofs can be modulo any combination of associativity and/or commu-
tativity and/or identity axioms; (3) identity axioms are automatically
transformed into rules by the semantics-preserving theory transforma-
tion in [4]; (4) support for user-specified recursive path orders modulo
any combination of the above axioms; and (5) support for user-specified
linear polynomial orders modulo any combination of the above axioms.
Keywords: RPO termination modulo, polynomial termination modulo,
term rewriting, Maude.

1 Introduction

Termination proofs are required for many formal verification purposes. Maude
has automated support for termination proofs thanks to its MTT tool [2]. MTT
works as follows: (1) First, the input module/theory is transformed in suitable
ways [3] to deal with issues such as subsorts, memberships, equations versus
rules, and conditional equations/rules. Non-termination-preserving transforma-
tions are applied that guarantee termination of the original specification if the
transformed one is terminating. (2) Then, termination tools such as AProVE
[5] or MU-TERM [1] are invoked as backends. These backend state-of-the-art
tools are fully automated and use sophisticated dependency pair techniques [10].
But, since termination is undecidable and the tools have timeouts and other
restrictions, some specifications cannot be proved terminating this way.

At present, if that happens (e.g., for IDEMPOTENT-SEMIGROUP in Section 3,
or for the example in Appendix A), no further assistance can be provided by
Maude tools: the user has to find other ways of proving the specification ter-
minating. But since some of the best automated tools have already been tried,
more direct user intervention is often needed. Could some tool assistance be pro-
vided? And if so, what should such assistance be like? An almost sine qua non
requirement is support for termination proofs modulo any combination of asso-
ciativity and/or commutativity and/or identity axioms: such axioms are heavily
used in many Maude specifications, so that a tool not supporting them would be
of marginal interest. Another requirement in a more interactive mode is placing

limited demands on the user’s expertise about termination techniques. For ex-
ample, dependency pairs and their termination proofs are quite complex, and it
would be hard for an average user to use them. A third desideratum is support for
conditional rules, which are used quite often in Maude specifications. The MTA
is of course not the most powerful such tool possible; but it meets these three
requirements reasonably well: (i) it supports termination proofs of order-sorted
Maude functional modules or theories modulo any combination of associativity
and/or commutativity and/or identity axioms; (ii) it does so for two well-known
termination proof methods that are quite easy to use, yet, remarkably effective
in proving termination for many examples, namely, RPO termination orders
modulo the above axioms, and polynomial termination orders also modulo the
above axioms; and (iii) supports proofs of conditional termination in the most
common case, which is also the easiest for a user to handle, namely, rules without
extra variables in their righthand sides or conditions. One additional advantage
—which actually stimulated us to develop MTA— is its usefulness for teaching
termination techniques to students in formal verification courses. Since RPO
and polynomial termination methods are among the most basic, a first intro-
duction to rewriting termination techniques should cover them. But there are
two practical problems: first, without some suitable tool no hads-on experience
can be provided to students; second, since students, at least in the courses of
this kind provided at the University of Illinois, become familiar with Maude and
its rewriting modulo axioms, such experience should allow them to prove ter-
minating specifications that use such axioms. The teaching experience with two
prototype versions of MTA in the last two years has been quite encouraging and
has motivated the present, more mature, general, and user-friendly version.

2 Theoretical Foundations

Termination Modulo A_C_U Axioms. The Maude specifications handled
by MTA are functional modules or theories of the form fmod pΣ,E ZBq endfm,

or fth pΣ,EZBq endfth, which are regarded as rewrite theories pΣ,B, ~Eq with
the equations E oriented as rewrite rules, and where the axioms B can be any
combination of associativity (A), commutativity (C) or identity (U) axioms. E
may contain conditional equations of the form t “ t1 if

Ź

i ui “ vi, but such that
varspt1qYvarsp

Ź

i ui “ viq Ď varsptq. However, conditional rules are transformed
into unconditional ones as explained below. Therefore, we need only focus on
explaining the kinds of unconditional termination proofs supported by MTA. All
of them are based on using a reduction ordering ą [10] on terms such that t ą t1

for each rule tÑ t1 in ~E, where ą is compatible with the axioms B. This exactly
means that the termination order is also defined on B-equivalence classes as,
rtsB ą rt

1sB . Identity axioms U are well-known to be problematic for reduction

orders. However, using the transformation in [4], the rewrite theory pΣ,B, ~Eq

can be transformed into a semantically equivalent one of the form pΣ,B0,
~̂
EY ~Uq,

where B “ B0 Z U is a decomposition into A_C-axioms B0 and U -axioms U ,
which are now oriented as rules ~U . This tranformation is applied at the outset

by MTA, so that the reduction orders used need only be A_C-compatible. Two
kinds of orders can be defined by an MTA user: A_C-compatible RPO orders,
or an A_C-compatible linear polynomial orders.

A_C RPO. RPO is a well-know simplification order parametric on an order
f ą g on function symbols [10]. Various versions of an AC-compatible RPO
order are well-known. What is perhaps less well-known, and is used crucially by
MTA, is that for any combination B0 of A and/or C axioms, and an ordering
on symbols in a finite Σ, a B0-compatible RPO order can be defined, which
furthermore becomes a total order on ground terms modulo B0, i.e., on TΣ{B0

.
The definition of all such A_C-compatible RPO orders and the proof of all these
properties is one of the important contributions of A. Rubio’s Ph.D. thesis [8].

Linear Polynomial Termination. For B0 any A_C axioms, call pΣ,B0, ~Eq
N-polynomially terminating iff each f P Σ can be associated a polynomial pf
with natural coefficients and exactly as many distinct variables as arguments in
f such that there exists a bound b P N, b ě 2, such that: (i) for all constants a
in Σ, pa ě b, (ii) for each u “ v in E pu ą pv as multi-argument functions on
Něb, where pt is the homomorphic extension to a term t of the map f ÞÑ pf ,
and (iii) if f P Σ satisfies some A and/or C axiom in B0, then pf does so too
as a polynomial function on Něb. Under such conditions the assignment f ÞÑ pf
defines a B0-compatible reduction order (see [10] for the case B0 “ H). There is,
however, one problem blocking full automation and needing heuristics, namely,
that given pΣ,B0, ~Eq it is in general undecidable whether a polynomial order of
this kind exists or not. This is because the problem can be reduced to a general
one of Diophantine equation solvability (see, e.g., [10]), which is well-known
to be undecidable by Matijasevich’s Theorem on Hilbert’s 10th Problem. MTA
adopts a simple approach to such an automation problem. First of all note that
a terminating N-polynomial order exists iff the formula

pDzq
ľ

aPΣ

z ě pa ^ p@xq
ľ

i

xi ě z ñ
ľ

u“vPE

pu ą pv

is valid in N, where x “ x1 . . . xk are the variables appearing in the equations E,
and we assume that different equations in E have disjoint variables. But if the
pf are all linear polynomials, then the above formula is in Presburger arithmetic
and its validity is therefore decidable. MTA does not use a decision procedure for
Presburger arithmetic (which in this case would require quantifier elimination).
It uses instead a simple special-purpose decision procedure written in Maude
to decide validity of the above formula for the given polynomial interpretation.
Finally, A_C-compatibility is easy to check for linear polynomials (more later).

Termination of Conditional Rules. Conditional equations t “ t1 if
Ź

i ui “
vi with the above-mention requirement on variables and oriented as rules with
their conditions interpreted as joinability conditions can be proved terminating
modulo B0 if there is a B0-compatible simplification order ą such that t ą t1,
t ą ui and t ą vi for each i [7]. This is achieved in MTA by: (i) transforming the
above rule into the rules t Ñ t1, t Ñ ui and t Ñ vi for each i, and (ii) ensuring
ą is a B0-compatible simplification order, which holds automatically for A_C



4 R. Gutierrez and J. Meseguer and S. Skeirik

RPO, and does hold for linear polynomial orders if for each unary f , pf “ ax`b
is such that either a ą 1 or if a “ 1 then b ě 1. MTA checks this last condition
if a polynomial order is used for a module with conditional equations.

3 MTA Syntax, Interaction, and Implementation

The syntax required for using the MTA is quite simple. It has two aspects: (i)
metadata notation (more later) for a user to specify for each function symbol or
constant in module or theory, say FOO, the information needed to define either
an A_C RPO order or a linear polynomial order; and (ii) the respective check
command syntax to be given after FOO and the tool have been loaded into Maude.

The user interaction is also quite simple. The user: (i) loads his/her module
or theory FOO in Maude with the desired meta-data annotations (we assume FOO

makes no use of built-in modules); (ii) loads the file mta.maude containing the
MTA tool implementation and providing a simple user interface as a Full Maude
extension; (iii) types the command (check-AvCrpo FOO .) if the defined order
is A_C RPO, or (check-poly FOO .) if it is a linear polynomial order; and (iv)
then receives output from the tool, either confirming that the termination check
was successful or listing the equations that could not be proved terminating. If
the check is unsuccessful, the user can easily modify the given order by changing
some metadata annotations or try a different kind of order.

Proving A_C RPO Termination. Consider the following module specifying
both lists and multisets and a l2m function transforming a list into a multi-
set. To avoid default inclusion of the BOOL built-in module, note the command
set include BOOL off ., which should always precede any module.

set include BOOL off .

fmod LIST+MSET is

sorts Element List MSet .

subsorts Element < List . subsorts Element < MSet .

op a : -> Element [ctor metadata "1"] .

op b : -> Element [ctor metadata "2"] .

op c : -> Element [ctor metadata "3"] .

op nil : -> List [ctor metadata "4"] .

op _;_ : List List -> List [metadata "5 lex(2 1)"] .

op _;_ : List Element -> List [ctor metadata "5 lex(2 1)"] .

op _,_ : MSet MSet -> MSet [ctor assoc comm metadata "4"] .

op null : -> MSet [ctor metadata "3"] .

op l2m : List -> MSet [ctor metadata "5"] .

vars L P Q : List . var M : MSet . var E : Element .

eq L ; (P ; Q) = (L ; P) ; Q . eq L ; nil = L .

eq nil ; L = L . eq M , null = M .

eq l2m(nil) = null . eq l2m(E) = E .

eq l2m(L ; E) = l2m(L) , E .

endfm

An A_C RPO order is here defined by a function level : Σ Ñ N and
each symbol’s lexicographic status, if any. level defines an order on symbols by

f ą g ô levelpfq ą levelpgq. For each f P Σ, levelpfq “ k is specified by adding
to f the declaration: metadata “k”. If f does not satisfy any A_C axioms, a lexi-
cographic status [10] can be specified by adding: lexpi1 . . . inq, where pi1 . . . inq is
a permutation of f ’s n arguments, as done above for the ; symbol. After load-
ing LIST+MSET and mta.maude, the command (check-AvCrpo LIST+MSET .) is
answered with output: Module is terminating by AvC-RPO order.

Proving A_C Polynomial Termination. The theory of idempotent semi-
groups is conjectured impossible to specify by unconditional confluent and ter-
minating rules in [9]. The authors give a locally confluent (modulo A_C) con-
ditional specification expressible in Maude as the following functional theory :

set include BOOL off .

fth IDEMPOTENT-SEMIGROUP is

sorts Semigroup Set .

op _ _ : Semigroup Semigroup -> Semigroup [assoc metadata "1 1 10"] .

op _,_ : Set Set -> Set [assoc comm metadata "1 1 0"] .

op {_} : Semigroup -> Set [metadata "1 1"].

var S : Set . vars L P Q : Semigroup .

eq L L = L .

ceq L P Q = L Q if {L} = {Q} /\ {L P} = {L} .

eq S , S = S .

eq {L P} = {L} , {P} .

endfth

The key idea is to use the function {_} to turn semigroup elements into set ele-
ments and use it in the second equation’s condition, so that the original idempo-
tency equation plus the added conditional equation make the theory locally con-
fluent. To show that the IDEMPOTENT-SEMIGROUP specification makes the theory
decidable we need to prove its operational termination. This can be done in MTA
using a linear polynomial order. The map p : f ÞÑ pf “ a1x1 ` . . . anxn ` an`1

for each f P Σ with n arguments is specified in Maude by declaring the operator
f with the metadata attribute: metadata “a1 . . . an an`1” subject to the follow-
ing requirements: (i) all a1, . . . , an are nonzero, (ii) for a constant c we must
have pc “ a1 ě 2, (iii) for f a binary symbol with pf “ a1x1 ` a2x2 ` a3, if f
is commutative, pf must be a symmetric polynomial, so that a1 “ a2, and if f
is associative (with or without commutativity) we must have a1 “ a2 “ 1. Fur-
thermore, if, as in this case, some equations are conditional, to make the order
a simplification order [10], (iv) if f is unary with pf “ a1x1` a2 we must either
have a1 ą 1 or if a1 “ 1 then a2 ě 1. Note that all these requirements are met by
the above meta-data annotations. After loading into Maude the above theory and
mta.maude and giving the command (check-poly IDEMPOTENT-SEMIGROUP .)

we get the output: Module is terminating by polynomial order.

Reflective Maude Implementation. Since MTA manipulates terms and the-
ories, its most natural implementation is a reflective one extending Maude’s
META-LEVEL, where terms and theories are available as data elements of respec-
tive sorts Term and Module. The MTA implementation extends Full Maude and

has three components: (i) an A_C RPO library, which is of independent inter-
est for resolution theorem proving modulo A_C, e.g., [8], or for Knuth-Bendix
completion modulo A_C, (ii) a decision procedure for linear polynomial ter-
mination; and (iii) a simple user interface accepting user commands to check
termination and reporting answers and errors. The MTA implementation with
some examples can be found in http://maude.cs.illinois.edu/tools/mta/.

4 Conclusions and Future Work

MTA complements the fully automatic MTT Tool [2] to assist users when MTT
cannot prove terminating a functional module or theory having A_C_U axioms
and possibly with conditional equations without extra variables in conditions or
righthand sides. It also complements AGES [6], a quite general tool that can
prove polynomial termination of conditional Maude modules with subsorts, but
without axioms. MTA can also be used as a teaching tool to give students a
hands-on experience in defining RPO or polynomial orders modulo axioms. Af-
ter more experimentation, MTA could be extended in several ways, including: (i)
support for termination proofs of Maude system modules and theories; (ii) sup-
port for polynomial termination proofs using non-linear polynomials; and (iii)
integration with MTT and other tools within the Maude Formal Environment.

A An Example

As mentioned in the Introduction, the IDEMPOTENT-SEMIGROUP theory in Section
3 cannot be proved terminating by MTT. This is because MTT does not han-
dle associative but non-commutative axioms. Likewise, MTT cannot prove the
ABSTRACT-BAKERY module below terminating, but MTA can prove its A_C-RPO
termination (in this second case MTT seems to have no problem with axioms).

Abstract Bakery. This functional module specifies an equational abstraction
for a two-process version of Lamport’s Bakery Mutual Exclusion Protocol. Bak-
ery is an infinite-state system; so it cannot be model checked by explicit-state
techniques. The ABSTRACT-BAKERY functional module below:

set include BOOL off .

fmod NAT> is

sorts Zero NzNat Nat Bool .

subsorts Zero NzNat < Nat .

op tt : -> Bool [ctor metadata "1"] . *** true

op ff : -> Bool [ctor metadata "2"] . *** false

op 0 : -> Zero [ctor metadata "3"] .

op 1 : -> NzNat [ctor metadata "4"] .

op _+_ : Nat Nat -> Nat [ctor assoc comm id: 0 metadata "12"] .

op _+_ : NzNat Nat -> NzNat [ctor assoc comm id: 0 metadata "12"] .

op _+_ : Nat NzNat -> NzNat [ctor assoc comm id: 0 metadata "12"] .

op _+_ : NzNat NzNat -> NzNat [ctor assoc comm id: 0 metadata "12"] .

op _>_ : Nat Nat -> Bool [metadata "13"] .

vars n m : Nat . vars n’ m’ : NzNat .

eq n + n’ > n = tt .

eq n > n + m = ff .

endfm

fmod BAKERY is

protecting NAT> .

sorts Mode BState .

op sleep : -> Mode [ctor metadata "5"] .

op wait : -> Mode [ctor metadata "6"] .

op crit : -> Mode [ctor metadata "7"] .

op <_,_,_,_> : Mode Nat Mode Nat -> BState [ctor metadata "14"] .

endfm

fmod BAKERY-PREDS is

protecting BAKERY .

sort State Prop .

subsort BState < State .

ops 1wait : -> Prop [ctor metadata "8"] .

ops 2wait : -> Prop [ctor metadata "9"] .

ops 1crit : -> Prop [ctor metadata "10"] .

ops 2crit : -> Prop [ctor metadata "11"] .

op _|=_ : State Prop -> Bool [frozen ctor metadata "15"] .

vars P Q : Mode .

vars X Y : Nat .

eq < wait, X, Q, Y > |= 1wait = tt .

eq < sleep, X, Q, Y > |= 1wait = ff .

eq < crit, X, Q, Y > |= 1wait = ff .

eq < P , X, wait, Y > |= 2wait = tt .

eq < P , X, sleep, Y > |= 2wait = ff .

eq < P , X, crit, Y > |= 2wait = ff .

eq < crit , X, Q, Y > |= 1crit = tt .

eq < sleep, X, Q, Y > |= 1crit = ff .

eq < wait, X, Q, Y > |= 1crit = ff .

eq < P , X, crit, Y > |= 2crit = tt .

eq < P , X, sleep, Y > |= 2crit = ff .

eq < P , X, wait, Y > |= 2crit = ff .

endfm

fmod ABSTRACT-BAKERY is

including BAKERY-PREDS .

vars P Q : Mode . vars X Y : Nat . vars X’ Y’ : NzNat .

eq < P, 0, Q, 1 + Y’ > = < P, 0, Q, 1 > .

eq < P, 1 + X’, Q, 0 > = < P, 1 , Q, 0 > .

eq < P, 1, Q, 1 + Y’ + X’ > = < P, 1, Q, 1 + 1 > .

eq < P, 1 + X’ + Y’, Q, 1 > = < P, 1 + 1, Q, 1 > .

eq < P, X’ + 1, Q, Y’ + X’ + 1 > = < P, 1, Q, 1 + 1 > .

eq < P, X’ + Y’ + 1, Q, Y’ + 1 > = < P, 1 + 1, Q, 1 > .

eq < P, X’ + Y’ + 1, Q, X’ + Y’ + 1 > = < P, 1 + 1, Q, 1 + 1 > .

endfm

defines the semantics of state predicates using the equations in the BAKERY-PREDS
submodule, and uses the last seven equations to identify system states (4-tuples
with ticket number and a mode for each process) so as to make the system
finite-state. This enables LTL model checking (with rewrite rules for system
transitions in a Maude system module not shown here, and with the above-
defined state predicates). The correctness of the associated LTL verification de-
pends on checking several crucial proof obligations, including local confluence
(done using Maude’s CRC Tool), and termination of ABSTRACT-BAKERY. How-
ever, MTT does not succeed in finding a termination proof. But MTA can prove
ABSTRACT-BAKERY terminating: Module is terminating by AvC-RPO order.

Note the somewhat interesting fact that the axioms for ` in submodule NAT>
include the identity axiom, which is outside the scope of A_C-RPO orders.
Before A_C-RPO termination is checked for ABSTRACT-BAKERY, the module
is transformed by MTA into a semantically equivalent module where such an
identity axioms has been transformed into the equation n`0 “ n and “variants”
of all other equations using the rule n` 0 Ñ n modulo AC have been computed
according to the semantics-preserving transformation described in [4].

